
TITLE: 
PMAT04: motion of projectiles. 
 
PROBLEM: 
We want to study the motion of a projectile in vacuum, subjected to the gravity force. The initial 
velocity is given and has a value of 100 m/s, however the shooting angle may vary. 
 
MESH: 
The model is 2D and uses three elements of type PMAT to represent three projectiles which are shot at 
different initial angles: 30°, 45° and 60°, respectively. 
The mesh includes also 3000 elements of type FUNE (2-noded bars) which, however, are only used to 
represent the analytical trajectories of the three projectiles. This allows to visually compare, at each 
time instant, the numerical and analytical positions. 
 
MATERIALS: 
The projectiles use a material of type MASS (concentrated mass), while the auxiliary FUNE elements 
used to visualize the trajectories are associated with a FANT material (phantom) and thus do not 
intervene in any way in the calculation. 
 
INITIAL CONDITIONS: 
The three projectiles have the same initial velocity in modulus, but different shooting angles. 
 
LOADING: 
A standard gravity load is applied to the three projectiles by means of the CHAR CONS GRAV 
directive. 
 
CALCULATION: 
The calculation is performed up to 17.7 s over 1000 time steps of fixed length. At the final time, the 
third projectile has reached the ground. 
RESULTS: 
Results are in perfect agreement with the analytical solutions. 

POST-TREATMENT 
An animation of the computed results from this calculation is made. 
 
Numerical Solutions 
 
PMAT04 
 
The mesh generation file (K2000): 
 
*%siz 100 
opti echo 1; 
* 
opti titr 'PMAT - 04'; 
opti dime 2 elem qua4; 
opti trac psc; 
opti ftra 'pmat04_mesh.ps'; 
* 
p0=0 0; 
g = 9.80665d0; 
v0 = 100.; 
n = 1000; 
* 
* First projectile 
* 
phi0 = 30.d0; 
sinp = sin phi0; 
cosp = cos phi0; 
tfin = 2.d0 * v0 * sinp / g; 
dt = tfin / n; 
* 
par1 = manu poi1 (p0 plus (0 0)); 
* 
x y = coor p0; 
t = 0.0; 
i = 0; 
p2 = p0 plus (0 0); 
repe loop1 n; 
 i = i + 1; 
 t = t + dt; 
 x = v0 * t * cosp; 
 y = v0 * t * sinp - (0.5d0 * g * t * t); 
 p1 = p2; 
 p2 = x y; 
 ele = manu seg2 p1 p2; 
 si (i ega 1); 
   tra1 = ele; 
 sinon; 
   tra1 = tra1 et ele; 
 finsi; 
fin loop1; 
* 
* Second projectile 
* 
phi0 = 45.d0; 
sinp = sin phi0; 
cosp = cos phi0; 
tfin = 2.d0 * v0 * sinp / g; 
dt = tfin / n; 
* 
par2 = manu poi1 (p0 plus (0 0)); 
* 
x y = coor p0; 
t = 0.0; 
i = 0; 
p2 = p0 plus (0 0); 

repe loop2 n; 
 i = i + 1; 
 t = t + dt; 
 x = v0 * t * cosp; 
 y = v0 * t * sinp - (0.5d0 * g * t * t); 
 p1 = p2; 
 p2 = x y; 
 ele = manu seg2 p1 p2; 
 si (i ega 1); 
   tra2 = ele; 
 sinon; 
   tra2 = tra2 et ele; 
 finsi; 
fin loop2; 
* 
* Third projectile 
* 
phi0 = 60.d0; 
sinp = sin phi0; 
cosp = cos phi0; 
tfin = 2.d0 * v0 * sinp / g; 
dt = tfin / n; 
* 
par3 = manu poi1 (p0 plus (0 0)); 
* 
x y = coor p0; 
t = 0.0; 
i = 0; 
p2 = p0 plus (0 0); 
repe loop3 n; 
 i = i + 1; 
 t = t + dt; 
 x = v0 * t * cosp; 
 y = v0 * t * sinp - (0.5d0 * g * t * t); 
 p1 = p2; 
 p2 = x y; 
 ele = manu seg2 p1 p2; 
 si (i ega 1); 
   tra3 = ele; 
 sinon; 
   tra3 = tra3 et ele; 
 finsi; 
fin loop3; 
* 
mesh=par1 et par2 et par3 et tra1 et tra2 et tra3; 
* 
tass mesh; 
trac qual mesh; 
* 
opti sauv form 'pmat04.msh'; 
sauv form mesh; 
* 
opti trac mif; 
trac tra2; 
fin; 

 

The EUROPLEXUS input file is: 
 
PMAT - 04 
ECHO 
 CONV win 
CPLA LAGR 
CAST MESH 
DIME 
  PT2L 3006 PMAT 3 FUN2 3000 ZONE 2 
TERM 
GEOM PMAT par1 par2 par3 FUN2 tra1 tra2 tra3 TERM 
COMP EPAI 10.0 LECT par1 par2 par3 TERM 
           1.0 LECT tra1 tra2 tra3 TERM 
     COUL roug LECT par1 TERM 
          jaun LECT par2 TERM 
          rose LECT par3 TERM 
          vert LECT tra1 TERM 
          bleu LECT tra2 TERM 
          turq LECT tra3 TERM 
MATE FANT 1.0 LECT tra1 tra2 tra3 TERM 
     MASS 1.0 LECT par1 par2 par3 TERM 
INIT VITE 1 86.6025404D0 LECT par1 TERM 
     VITE 2 50.0000000D0 LECT par1 TERM 
     VITE 1 70.7106781D0 LECT par2 TERM 
     VITE 2 70.7106781D0 LECT par2 TERM 
     VITE 1 50.0000000D0 LECT par3 TERM 
     VITE 2 86.6025404D0 LECT par3 TERM 
CHAR CONST GRAV 0 -9.80665D0 LECT par1 par2 par3 TERM 
ECRI DEPL VITE FREQ 100 
          POIN LECT par1 par2 par3 TERM 
     FICH ALIC TEMP FREQ 1 
               POIN LECT par1 par2 par3 TERM 
     FICH ALIC FREQ 250 
OPTI NOTE PAS UTIL 
 log 1 
CALCUL TINI 0. TEND 17.66200291D0 PASF 17.66200291E-3 NMAX 1000 
*================================================================= 
PLAY 
CAME   1 EYE   5.09858E+02  1.27465E+02  2.62775E+03 
!        Q     1.00000E+00  0.00000E+00  0.00000E+00  0.00000E+00 
         VIEW  0.00000E+00  0.00000E+00 -1.00000E+00 
         RIGH  1.00000E+00  0.00000E+00  0.00000E+00 
         UP    0.00000E+00  1.00000E+00  0.00000E+00 
         FOV   2.68819E+01 
SCEN GEOM NAVI FREE 
          LINE HEOU 
          POIN SPHP 
     COLO PAPE 
     VECT SCCO FIEL VITE SCAL USER PROG 47 PAS 4 99 TERM 
     TEXT VSCA 
SLER CAM1 1 NFRA 1 
FREQ 10 
TRAC OFFS FICH AVI NOCL NFTO 101 FPS 15 KFRE 10 COMP -1 REND 
GOTR LOOP 99 OFFS FICH AVI CONT NOCL REND 
GO 
TRAC OFFS FICH AVI CONT REND 
ENDPLAY 
*================================================================= 
SUIT 
Post-treatment 
ECHO 
* 
RESU ALIC TEMP GARD PSCR 

* 
SORT GRAP 
* 
AXTE 1.0 'Time [s]'  
* 
COUR  1 'dx_par1' DEPL COMP 1 NOEU LECT par1 TERM 
COUR  2 'dy_par1' DEPL COMP 2 NOEU LECT par1 TERM 
COUR  3 'dx_par2' DEPL COMP 1 NOEU LECT par2 TERM 
COUR  4 'dy_par2' DEPL COMP 2 NOEU LECT par2 TERM 
COUR  5 'dx_par3' DEPL COMP 1 NOEU LECT par3 TERM 
COUR  6 'dy_par3' DEPL COMP 2 NOEU LECT par3 TERM 
* 
trac 1 2 3 4 5 6 axes 1.0 'DISPL. [M]' yzer 
     COLO vert vert bleu bleu turq turq 
trac   2   4   6 axes 1.0 'DISPL. [M]' yzer 
     COLO      vert      bleu      turq 
trac 2 axes 1.0 'Y-DISPL. [M]' xaxe 1 1.0 'X-DISPL. [M]' yzer 
     COLO vert 
trac 4 axes 1.0 'Y-DISPL. [M]' xaxe 3 1.0 'X-DISPL. [M]' yzer 
     COLO bleu 
trac 6 axes 1.0 'Y-DISPL. [M]' xaxe 5 1.0 'X-DISPL. [M]' yzer 
     COLO turq 
list 2 axes 1.0 'Y-DISPL. [M]' xaxe 1 1.0 'X-DISPL. [M]' 
list 4 axes 1.0 'Y-DISPL. [M]' xaxe 3 1.0 'X-DISPL. [M]' 
list 6 axes 1.0 'Y-DISPL. [M]' xaxe 5 1.0 'X-DISPL. [M]' 
* 
QUAL DEPL COMP 1 LECT par3 TERM REFE 8.831001451D+2 TOLE 1.E-6 
     DEPL COMP 2 LECT par3 TERM REFE 0.000000000D+0 TOLE 1.E-6 
*================================================================= 
SUIT 
Post-treatment (rendering on bitmap file) 
ECHO 
* 
RESU ALIC GARD PSCR 
* 
SORT VISU NSTO 1 
* 
*================================================================= 
PLAY 
CAME   1 EYE   5.09858E+02  1.27465E+02  2.62775E+03 
!        Q     1.00000E+00  0.00000E+00  0.00000E+00  0.00000E+00 
         VIEW  0.00000E+00  0.00000E+00 -1.00000E+00 
         RIGH  1.00000E+00  0.00000E+00  0.00000E+00 
         UP    0.00000E+00  1.00000E+00  0.00000E+00 
         FOV   2.68819E+01 
SCEN GEOM NAVI FREE 
          LINE HEOU 
          POIN SPHP 
     COLO PAPE 
     VECT SCCO FIEL VITE SCAL USER PROG 47 PAS 4 99 TERM 
     TEXT VSCA 
SLER CAM1 1 NFRA 1 
FREQ 1 
TRAC OFFS FICH BMP REND 
GOTR LOOP 3 OFFS FICH BMP REND 
GO 
TRAC OFFS FICH BMP REND 
ENDPLAY 
*================================================================= 
FIN 

Some results: horizontal and vertical displacement components as a function of time 
(note that motion is uniform in the horizontal direction since there is no air resistance 
in the modelled case): 
 

 
 
Vertical displacement components as a function of time: 

Trajectories: 
 

 
 
These results are in perfect agreement with the analytical results. An animation of the 
motion allows to visually compare the computed positions with the analytical ones: 

PMAT01E 
 
This test is similar to PMAT04 but uses only one projectile (the one at 30°) and just 
one time step to compute the final position of the projectile as it hits the ground. 
 
The solution is in perfect agreement with the analytical one. This should not be 
surprising, since: 

• In this problem there are no deformable structures, and therefore no wave 
propagation phenomena. Consequently, the usual stability requirements 
(Courant) of explicit time integration methods do not hold in the present case 
and one may use an arbitrarily large time increment (as far as stability is 
concerned). 

• As concerns accuracy, the central difference scheme is second-order accurate 
and therefore it reproduces exactly the analytical solution which, in this case, 
is a second-degree polynomial (parabola). 

 
A summary of solutions computed with different time increments is given in the 
following Table: 
 

 
 


